Fertility And Development

Written by Mark Lauer, August 30th, 2009
Last updated September 3rd, 2009

This notebook generates graphics from the data used in the paper:
Mikko Myrskylä, Hans-Peter Kohler & Francesco C. Billari (2009)
"Advances in development reverse fertility declines"
Nature 460, 741-743 (6 August 2009) | doi:10.1038/nature08230
http://www.nature.com/nature/journal/v460/n7256/full/nature08230.html

Import Data

Download and import the data

In[1]:=
```
dataurl = "http://www.nature.com/nature/journal/v460/n7256/extref/nature08230-s2.zip";
data = First[Import[dataurl, "*"]];
TableForm[data[[Range[5], Range[10]]]]
```

Out[2]//TableForm=

country	HDI.1975	HDI.1976	HDI.1977	HDI.1978	HDI.1979	HDI.1980	HDI.1981	HDI.1982	HDI.1983
Albania						0.731273	0.734932	0.737146	0.738452
Algeria	0.565067	0.570966	0.575103	0.581995	0.587976	0.590365	0.593946	0.599331	0.605605
Angola	0.427032	0.427252	0.427463	0.428155	0.428839	0.429516	0.424447	0.422965	0.423151
Argentina	0.796896	0.796294	0.800532	0.798749	0.804949	0.807989	0.805496	0.80343	0.805606

Extract and remove the list of countries and column headings, then report the length of each

In[3]:=
```
countrylist = Rest[data[[All, 1]]];
headinglist = Rest[data[[1]]];
data = Drop[Transpose[Drop[Transpose[data], 1]], 1];
TableForm[{{"Countries: ", Length[countrylist]}, {"Columns: ", Length[headinglist]}}]
```

Out[6]//TableForm=

Countries:	143
Columns:	124

A function to parse column headings and define corresponding *Mathematica* functions from the values.
For example, "HDI.1975" leads to defining HDI[countryname, 1975]

In[7]:=
```
Store[value_, {country_Integer, columnname_Integer}] :=
 Module[{type, year},
  {type, year} = StringSplit[headinglist[[columnname]], "."];
  (Symbol[type][countrylist[[country]], ToExpression[year]]) =
   If[NumberQ[value], value, Missing[]]
 ]
```

Apply this across all the data

In[8]:=
```
MapIndexed[Store, data, {2}];
```

Check this with a couple of plots

In[9]:=
```
ListPlot[Table[{y, HDI["Australia", y]}, {y, 1975, 2005}]]
```

Out[9]=



In[10]:=
```
ListPlot[Table[{y, TFR["Australia", y]}, {y, 1975, 2005}]]
```

Out[10]=



Generate a scatter plot of all countries' TFR against HDI in 1975 and 2005

In[11]:=
```
ListPlot[{
  {HDI[#, 2005], TFR[#, 2005]} & /@ countrylist,
  {HDI[#, 1975], TFR[#, 1975]} & /@ countrylist}
]
```

Out[11]=



Generate a similar plot, but with distorted axes, as the paper does, to make differences at low fertility and high development look much more significant

In[12]:=
```
ListLogPlot[{
  {-Log[1 - HDI[#, 2005]], TFR[#, 2005]} & /@ countrylist,
  {-Log[1 - HDI[#, 1975]], TFR[#, 1975]} & /@ countrylist} ,
 Ticks → {{-Log[1 - #], #} & /@ {0.3, 0.6, 0.8, 0.9, 0.95}, Automatic},
 PlotRange → {{-Log[1 - 0.25], 3.5}, {1, 9}},
 AspectRatio → 1
]
```

Out[12]=



Match Countries To *Mathematica* Country Data

Define an equivalent list of countries using *Mathematica* names by expanding some abbreviations and removing spaces

In[13]:=
```
canonicallist =
  (countrylist /. {"USA" → "UnitedStates", "Congo, Dem. Rep." → "DemocraticRepublicCongo",
     "Congo, Rep." → "RepublicCongo", "Cote d'Ivoire" → "IvoryCoast",
     "Kyrgyz Republic" → "Kyrgyzstan", "NL" → "Netherlands", "S. Korea" → "SouthKorea",
     "Slovak Republic" → "Slovakia", "Trinidad and Tobago" → "TrinidadTobago",
     "Lao" → "Laos", x_String :> StringReplace[x, {" " → ""}]});
```

Check that every country in the data matches one in *Mathematica*

In[14]:=
```
Complement[canonicallist, CountryData["Countries"]] == {}
```

Out[14]=
```
True
```

Key Functions

Define (self-cacheing) function to map countries to the *Mathematica* names using the list

In[15]:=
```
CanonicalName[country_String] :=
 (CanonicalName[country] = canonicallist[[First[First[Position[countrylist, country]]]]])
```

Check this for three countries

```
In[16]:= CanonicalName /@ {"USA", "New Zealand", "United Kingdom"}
Out[16]= {UnitedStates, NewZealand, UnitedKingdom}
```

Get list of continents for countries

```
In[17]:= continentslist = Union[CountryData[CanonicalName[#], "Continent"] & /@ countrylist]
Out[17]= {Africa, Asia, Europe, NorthAmerica, Oceania, SouthAmerica}
```

Define (self-cacheing) function to map countries to continents

```
In[18]:= ContinentOf[country_String] :=
          (ContinentOf[country] = CountryData[CanonicalName[country], "Continent"])
```

Check this for four countries

```
In[19]:= ContinentOf /@ {"USA", "China", "Israel", "Australia"}
Out[19]= {NorthAmerica, Asia, Asia, Oceania}
```

Define ColourOf[] function from continents to colours by splitting the (reversed) DarkRainbow spectrum, then display all values

```
In[20]:= MapThread[Set, {ColourOf /@ Reverse[continentslist],
            ColorData["DarkRainbow"] /@ (Range[Length[continentslist]] / Length[continentslist])}];
         Style[#, FontColor → ColourOf[#]] & /@ continentslist
Out[21]= {Africa, Asia, Europe, NorthAmerica, Oceania, SouthAmerica}
```

Define (self-cacheing) function to map countries to their populations according to *Mathematica*

```
In[22]:= PopulationOf[country_String] :=
          (PopulationOf[country] = CountryData[CanonicalName[country], "Population"])
```

Check this for four countries

```
In[23]:= PopulationOf /@ {"USA", "China", "Israel", "Australia"}
Out[23]= {3.02841 × 10^8, 1.29801 × 10^9, 6.80999 × 10^6, 2.05304 × 10^7}
```

Animated Charts

Define function to plot a bubble chart of Total Fertility Rate against Human Development Index for a given year. Bubble sizes are determined by population, colours by continent.

In[24]:=
```
SnapshotChart[year_Integer, chartoptions___] :=
 BubbleChart[{
   (* Ensure legend appears in fixed order by "plotting" continents *)
   Legended[Style[{0, 0, 1}, ColourOf[#]], Style[#, Small]] &
    /@ continentslist,
   (* Add bubble for each country *)
   Style[
      (* On mouse-over, display country names *)
      Tooltip[
       {HDI[#, year], TFR[#, year], PopulationOf[#]},
       #],
      ColourOf[ContinentOf[#]]
     ] & /@ countrylist},
  chartoptions,
  BubbleSizes → {0.01, 0.15}, PlotRange → {{0.2, 1.0}, {0.8, 9}},
  FrameLabel → {"Human Development Index", "Total Fertility Rate"},
  PlotLabel → ToString[year]
 ]
```

Check this for one year

In[27]:=
```
SnapshotChart[2002]
```

Out[27]=



Use *Mathematica*'s built in dynamic graphics to view animation through time

```
Manipulate[SnapshotChart[y], {y, 1975, 2005, 1}]
```

Generate an animated GIF of all thirty years
(Note: mouse-over will no longer work outside *Mathematica*)

```
Export["FertilityAndDevelopment.gif",
 Table[SnapshotChart[y], {y, 1975, 2005, 1}], ImageSize → 440]
```

```
FertilityAndDevelopment.gif
```

Generate an animated GIF of all thirty years, zooming in to region with advanced countries
(Note: mouse-over will no longer work outside *Mathematica*)

In[26]:=
```
Export["FertilityAndDevelopmentDetail.gif", Table[
  SnapshotChart[y, PlotRange → {{0.7, 1.0}, {0.8, 4}}],
  {y, 1975, 2005, 1}], ImageSize → 440]
```

Out[26]=
```
FertilityAndDevelopmentDetail.gif
```

Trajectory Plots

Use the threshold found by the paper as the HDI level at which TFR changes direction

```
bouncethreshold = 0.86;
```

Find countries which reach or exceed this threshold at some point in available data

```
advanced = Select[countrylist, (Max[Table[HDI[#, i], {i, 1975, 2005}]] ≥ bouncethreshold) &]
```

```
{Argentina, Australia, Austria, Belgium, Canada, Cyprus, Denmark,
 Finland, France, Germany, Greece, Hungary, Iceland, Ireland, Israel, Italy,
 Japan, S. Korea, Kuwait, Luxembourg, Malta, NL, New Zealand, Norway, Portugal,
 Spain, Sweden, Switzerland, United Arab Emirates, United Kingdom, USA}
```

Define a function to determine the first year in which a country's HDI exceeds a given threshold

```
ReferenceYear[country_String, developmentthreshold_Real] :=
 Min[
  Select[Range[1975, 2005],
   (HDI[country, #] ≥ developmentthreshold) &]
 ]
```

Check this function using the paper's threshold for all advanced countries

```
TableForm[SortBy[
  {#, ReferenceYear[#, bouncethreshold]} & /@ advanced,
  Last]]
```

```
Canada               1975
Denmark              1975
NL                   1975
Norway               1975
Sweden               1975
Switzerland          1975
USA                  1975
France               1976
Japan                1976
Australia            1978
Belgium              1978
Finland              1978
Iceland              1978
Austria              1980
Italy                1981
New Zealand          1982
Spain                1982
United Kingdom       1982
Germany              1983
Luxembourg           1984
Israel               1985
Ireland              1990
Greece               1992
Cyprus               1995
Portugal             1997
S. Korea             1997
Malta                2001
Kuwait               2003
United Arab Emirates 2004
Argentina            2005
Hungary              2005
```

Define a function to return the time series of TFR for a given country beginning in its reference year (measured as absolute difference from the TFR in the reference year).

```
TFRSeriesFromReference[country_String, threshold_Real] :=
 If[ReferenceYear[country, threshold] > 2005,
  (* Return empty series when HDI never reached threshold *)
  {},
  Table[TFR[country, y], {y, ReferenceYear[country, threshold], 2005}] -
   TFR[country, ReferenceYear[country, threshold]]
 ]
```

Check this function for Australia

```
ListPlot[TFRSeriesFromReference["Australia", bouncethreshold]]
```



Define a reasonable colour scheme

```
stylescheme =
  Reverse[ColorData["DarkRainbow"] /@ (Range[Length[advanced]] / Length[advanced])];
```

Build a chart of these time series for all advanced countries

```
g0 = ListPlot[
  Cases[
   TFRSeriesFromReference[#, bouncethreshold] & /@ advanced,
   (* Eliminate empty series *)
   {_, ___}],
  Joined → True,
  PlotStyle → stylescheme, PlotLabel →
   "Total Fertility Changes in Advanced Countries\nfrom First Year in which HDI reaches " <>
    ToString[bouncethreshold], AxesLabel → {"Years", "Change in TFR"}]
```



Generate a PNG file containing this chart

```
Export["FertilitySeries.png", g0, ImageSize → 420]
```

```
FertilitySeries.png
```

Define a function to return a pair of changes, the first being the absolute change in HDI, the second being the absolute change in TFR, between the reference year and 2005.

```
ChangeFromReference[country_String, threshold_Real] :=
 Module[{refyear = ReferenceYear[country, threshold]},
  If[refyear > 2005, {Infinity, Infinity},
   {HDI[country, 2005] - HDI[country, refyear],
    TFR[country, 2005] - TFR[country, refyear]}
  ]
 ]
```

Plot these changes in a scatter plot for all advanced countries

```
g1 =
 ListPlot[{ChangeFromReference[#, bouncethreshold]} & /@ advanced, PlotMarkers → advanced,
  PlotRange → {{-0.015, 0.11}, {-0.8, 0.4}}, PlotStyle → stylescheme]
```



The paper fits a model in which countries with HDI above the 0.86 threshold see increasing TFR with increasing HDI, according to which "on average an HDI increase of 0.05 results in an increase of the TFR by 0.204".
Build a plot illustrating this rate of increase for later addition to the plot above.

```
g2 = ListPlot[{{0, 0}, {0.1, 0.408}},
   Joined → True, PlotRange → All, PlotStyle → {{Red, Thick, Dashed}}];
```

Define a function to return a complete trajectory of changes in HDI and TFR between the reference year and 2005 (both measured relative to their value in the reference year).

```
TrajectoryFromReference[country_String, threshold_Real] :=
 Module[{refyear = ReferenceYear[country, threshold]},
  If[refyear > 2005,
   (* Keep non-advanced countries off plots *)
   {{Infinity, Infinity}},
   Table[
    {HDI[country, y] - HDI[country, refyear],
     TFR[country, y] - TFR[country, refyear]},
    {y, refyear, 2005}]
  ]
 ]
```

Build a plot which includes all these trajectories for advanced countries, then display these together with the scatter plot and fitted model plot above

```
(* Build trajectory plot *)
g3 = ListPlot[
   TrajectoryFromReference[#, bouncethreshold] & /@ advanced,
   PlotStyle → stylescheme,
   Joined → True];
(* Display all three graphs together with nice axes *)
g4 = Show[g1, g2, g3, PlotRange → {{-0.015, 0.112}, {-0.8, 0.4}},
  FrameLabel → {"Change in HDI", "Change in TFR"}, Axes → True,
  AxesStyle → GrayLevel[0.7], Frame → {{True, False}, {True, False}},
  PlotLabel → "Trajectories of Advanced Countries\nfrom First Year in which HDI reaches " <>
    ToString[bouncethreshold] <> "\n(dashed line is model fitted by the paper)"]
```



Generate a PNG file containing this chart

```
Export["FertilityTrajectories.png", g4, ImageSize → 420]
```

```
FertilityTrajectories.png
```